



명쾌한 기풍샘 함수 사전



함수 사전

1. 날짜와 시간함수

DATE, DAY, DAYS, EDATE, EOMONTH, HOUR, MINUTE, MONTH, NOW, SECOND, TIME, TODAY, WEEKDAY, WORKDAY, YEAR

- 1) NOW() : 현재 시스템의 날짜와 시간을 표시한다.
 - ▶ 괄호 안에 인수가 없다.
 - ▶ =NOW() → 2021-04-03 15:33
- 2) TODAY() : 현재 시스템의 날짜를 표시한다.
 - ▶ 괄호 안에 인수가 없다.
 - ▶ =TODAY() → 2021-04-03
- 3) YEAR(날짜) : 날짜에서 년도만 추출한다.
 - ▶ =YEAR("2021-04-03") → 2021
- 4) MONTH(날짜) : 날짜에서 월만 추출한다.
 - ▶ =MONTH("2021-04-03") → 4
- 5) DAY(날짜) : 날짜에서 일만 추출한다.
 - ▶ =DAY("2021-04-03") → 3
- 6) DATE(년,월,일) : 년, 월, 일을 조합하여 날짜를 표시한다.(날짜 일련번호)
 - ▶ =DATE(2021,4,3) → 2021-04-03(날짜 일련번호 : 44289)

날짜 일련번호란?

- ▶ 날짜 일련번호는 1900-1-1을 1로 하면 하루가 지날 때마다 1씩 증가한다.
- ▶ 현재 날짜를 표시하는 바로가기 키 : <Ctrl>+<:>

- 7) HOUR(시간) : 시간에서 시만 추출한다.
 - ▶ =HOUR("13:30") → 13
- 8) MINUTE(시간) : 시간에서 분만 추출한다.
 - ▶ =minute("13:30") → 30
- 9) SECOND(시간) : 시간에서 초만 추출한다.
 - ▶ =SECOND("13:30:45") → 45
- 10) TIME(시,분,초) : 시, 분, 초를 조합하여 시간을 표시한다.
 - ▶ =TIME(13,30,45) → 1:30 PM

시간

- ▶ 시간은 낮 12시가 숫자로 0.5이다.
- ▶ 현재 시간을 표시하는 바로가기 키 : <Ctrl>+<Shift>+<:>

- 11) DAYS(종료일,시작일) : 종료일에서 시작일을 뺀 일수를 구한다.
 - ▶ =DAYS("2021-4-30","2021-4-3") → 27
- 12) EDATE(날짜,K) : 날짜의 월에 K를 더한 날짜를 구한다.
 - ▶ =EDATE("2021-04-03",4) → 2021-08-03

- 13) EOMONTH(날짜,K) : 날짜의 월에 K를 더한 달의 마지막 날짜를 구한다.

- ▶ =EOMONTH("2021-04-03",7) → 2021-11-30

- 14) WEEKDAY(날짜,옵션) : 날짜의 요일 번호를 구한다.

- ▶ =WEEKDAY(날짜,1) : 1(일요일), 2(월요일) ~ 7(토요일)
- ▶ =WEEDDAY(날짜,2) : 1(월요일), 2(화요일) ~ 7(일요일)
- ▶ =WEEKDAY(날짜,3) : 0(월요일), 1(화요일) ~ 6(일요일)
- ▶ =WEEKDAY("2021-04-03,1) → 7(2021-04-03은 토요일이다.)
- ▶ =WEEKDAY("2021-04-03,2) → 6
- ▶ =WEEKDAY("2021-04-03,3) → 5

2. 논리함수

AND, OR, IF, IFERROR, NOT, TRUE, FALSE

- 1) =AND(조건1,조건2) : 조건1과 조건2가 모두 참(TRUE)이면 참(TRUE)이다.
 - ▶ 인수가 2개 이상 존재하면 모든 인수들이 조건으로 이루어져 있다.
 - ▶ =AND(3>2,4>3) → TRUE
 - ▶ =AND(3<2,4>3) → FALSE
- 2) =OR(조건1,조건2) : 조건1과 조건2 중 하나라도 참(TRUE)이면 참(TRUE)이다.
 - ▶ OR(3<2,4>3) → TRUE
 - ▶ OR(2>3,3>4) → FALSE
- 3) =IF(조건,참,거짓) : 조건을 만족하면 참(TRUE) 그렇지 않으면 거짓이다.
 - ▶ =IF(80>=70,"크다","작다") → 크다
- 4) IFERROR(값,오류값) : 값을 표시하되, 오류가 발생하면 오류값을 표시한다.
 - ▶ =IFERROR(3*4,4*5) → 12 : 첫 번째 인수가 정상적으로 계산이 되면 첫 번째 인수의 값을 표시한다.
 - ▶ =IFERROR(3/0,4*5) → 20 : 모든 수는 0으로 나눌 수 없어 오류가 발생하므로 두 번째 인수의 값이 표시된다.
- 5) =NOT(인수) : 인수가 TRUE이면 결과값은 FALSE를 표시하고 인수가 FALSE이면 결과값은 TRUE를 표시한다.
 - ▶ =NOT(FALSE) → TRUE
 - ▶ =NOT(1+1=2) → FALSE
 - ▶ =NOT(2+2=5) → TRUE

3. 데이터베이스함수

DSUM, DAVERAGE, DMAX, DMIN, DCOUNT, DCOUNTA, DVAR, DSTDEV



- 1) =DSUM(표범위, 열제목, 조건) : 표범위에서 조건을 만족하는 열제목의 합계를 표시한다.
- 2) =DAVERAGE(표범위, 열제목, 조건) : 표범위에서 조건을 만족하는 열제목의 평균을 표시한다.
- 3) =DMAX(표범위, 열제목, 조건) : 표범위에서 조건을 만족하는 열제목의 최대값을 표시한다.
- 4) =DMIN(표범위, 열제목, 조건) : 표범위에서 조건을 만족하는 열제목의 최소값을 표시한다.
- 5) =DCOUNT(표범위, 열제목, 조건) : 표범위에서 조건을 만족하는 열제목의 개수를 표시한다.(열제목 아래 데이터가 숫자, 날짜, 시간만 인식하면 문자는 인식하지 못한다.)
- 6) =DCOUNTA(표범위, 열제목, 조건) : 표범위에서 조건을 만족하는 열제목의 개수를 표시한다.(열제목 아래 데이터가 문자, 날짜, 시간, 숫자 모두 인식 가능하다.)
- 7) =DVAR(표범위, 열제목, 조건) : 표범위에서 조건을 만족하는 열제목의 분산을 표시한다.
- 8) =DSTDEV(표범위, 열제목, 조건) : 표범위에서 조건을 만족하는 열제목의 표준편차를 표시한다.

4. 문자열함수

LEFT, RIGHT, MID, LEN, LOWER, UPPER, PROPER, SEARCH, FIND, TRIM

- 1) =LEFT(문자, K) : 문자의 왼쪽에서 K번째까지 추출한다.
▶ =LEFT("교육의중심이비에스", 5) → 교육의중심
- 2) =RIGHT(문자, K) : 문자의 오른쪽에서 K번째까지 추출한다.
▶ =RIGHT("교육의중심이비에스", 4) → 이비에스
- 3) =MID(문자, 시작위치, K) : 문자를 시작위치에서 K번째까지 추출한다.
▶ =MID("교육의중심이비에스", 4, 2) → 중심
- 4) =LEN(문자) : 문자의 길이를 숫자로 표시한다.
▶ =LEN("교육의중심이비에스") → 9

공백이나 특수문자도 모두 한글자로 인식한다.

- 5) =LOWER(영어) : 영어를 소문자로 표시한다.
▶ =LOWER("I LOVE EBS") → i love ebs
- 6) =UPPER(영어) : 영어를 대문자로 표시한다.
▶ =UPPER("i love ebs") → I LOVE EBS
- 7) =PROPER(영어) : 영어 단어의 첫글자만 대문자로 표시한다.
▶ =PROPER("i love ebs") → I Love Ebs
- 8) =SEARCH("찾을 문자", "문자") : 문자에서 찾을 문자의 위치를 숫자로 표시한다.(대/소문자 구별하지 않는다.)
▶ =SEARCH("심", "교육의중심이비에스") → 5
▶ =SEARCH("E", "I Love Ebs") → 6 (공백도 1글자로 인식)
- 9) =FIND("찾을 문자", "문자") : 문자에서 찾을 문자의 위치를 숫자로 표시해 준다.(대/소문자 구별한다.)
▶ =FIND("심", "교육의중심이비에스") → 5
▶ =FIND("E", "I Love Ebs") → 8 (공백도 1글자로 인

식)

- 10) =TRIM(문자) : 문자열의 앞/뒤의 공백은 제거하고 나머지 공백은 한 칸으로 만든다.
▶ =TRIM(" 교육의 중심 EBS ") → 교육의 중심 EBS

5. 수학과 삼각함수

ABS, INT, MOD, POWER, ROUND, ROUNDUP, ROUNDDOWN, TRUNC, SUM, SUMIF, SUMIFS

- 1) =ABS(숫자) : 숫자를 절대값으로 표시한다.
▶ =ABS(-8) → 8
▶ =ABS(8) → 8
- 2) =INT(실수) : 실수를 가장 낮은 정수로 표시한다.
▶ =INT(3.9) → 3
▶ =INT(-3.9) → -4
- 3) =MOD(숫자, K) : 숫자를 K로 나눈 나머지를 표시한다.
▶ =MOD(12, 5) → 2
- 4) =POWER(숫자, K) : 숫자를 K만큼 거듭제곱해서 표시한다.
▶ =POWER(2, 3) → 8 : 2의 3승한 값이다.
- 5) =ROUND(숫자, 자릿수) : 숫자를 반올림하여 자릿수까지 표시한다.
▶ =ROUND(1728.1728, 3) → 1728.173 : 자릿수가 3이므로 소수 네 번째 자리에서 반올림하여 소수 세 번째 자리까지 표시한다.
▶ =ROUND(1728.1728, 0) → 1728 : 자릿수가 0이므로 소수 첫 번째 자리에서 반올림하여 정수로 표시한다.
▶ =ROUND(1728.1728, -1) → 1730 : 자릿수가 -1이므로 일의 자리에서 반올림하여 십의 자리까지 표시한다.
- 6) =ROUNDUP(숫자, 자릿수) : 숫자를 올림하여 자릿수까지 표시한다.
▶ =ROUNDUP(1728.1728, 2) → 1728.18 : 자릿수가 2이므로 소수 세 번째 자리에서 올림하여 소수 두 번째 자리까지 표시한다.
▶ =ROUNDUP(1728.1728, 0) → 1729 : 자릿수가 0이므로 소수 첫 번째 자리에서 올림하여 정수로 표시한다.
▶ =ROUNDUP(1728.1728, -2) → 1800 : 자릿수가 -2이므로 십의 자리에서 올림하여 백의 자리까지 표시한다.
- 7) =ROUNDDOWN(숫자, 자릿수) : 숫자를 내림하여 자릿수까지 표시한다.
▶ =ROUNDDOWN(1728.1728, 3) → 1728.172 : 자릿수가 3이므로 소수 네 번째 자리에서 내림하여 소수 세 번째 자리까지 표시한다.
▶ =ROUNDDOWN(1728.1728, 0) → 1728 : 자릿수가 0이므로 소수 첫 번째 자리에서 내림하여 정수로 표시한다.
▶ =ROUNDDOWN(1728.1728, -1) → 1720 : 자릿수가 -1이므로 일의 자리에서 내림하여 십의 자리까지 표시



한다.

- 8) =SUM(범위) : 범위 또는 인수들의 **합계**를 표시한다.
 ▶ =SUM(3,4,5) → 12
- 9) =SUMIF(조건범위,조건,합계범위) : **조건범위**에서 **조건**에 만족하는 값을 찾아서 **합계범위의 합계**를 구한다.
- 10) =SUMIFS(합계범위,조건범위1,조건1,조건범위2,조건2...) : **조건범위1**에서 **조건1**을 만족하고 **조건범위2**에서 **조건2**를 만족하는 **합계범위의 합계**를 구한다.

6. 찾기와 참조함수

CHOOSE, VLOOKUP, HLOOKUP, INDEX, MATCH

- 1) =CHOOSE(K,인수1,인수2,인수3...) : K값이 1이면 '인수1', K값이 2이면 '인수2', K값이 3이면 '인수3'을 표시한다.(K에 따라 뒤에 인수들이 결정된다.)
 ▶ =CHOOSE(2,"금","은","동") → 은 : K값이 2이므로 '은'이 표시된다.
- 2) =VLOOKUP(기준셀,참조범위,열번호,옵션) : **기준셀**에 참조되는 값을 **참조범위**에 있는 **열 번호**에서 찾아 표시한다.
 ▶ **옵션** : 정확한 값을 찾을 때는 0(FALSE), 유사한 값을 찾을 때는 1(TRUE)
 ▶ 기준셀이 문자이면 정확한 값 0을 입력하고 숫자이면 문제를 보고 0 또는 1을 입력한다.
- 3) =HLOOKUP(기준셀,참조범위,행번호,옵션) : **기준셀**에 참조되는 값을 **참조범위**에 있는 **행 번호**에서 찾아 표시한다.
 ▶ **옵션** : 정확한 값을 찾을 때는 0(FALSE), 유사한 값을 찾을 때는 1(TRUE)
 ▶ 기준셀이 문자이면 정확한 값 0을 입력하고 숫자이면 유사한 값 1을 입력한다.

	A	B	C	D	E	F	G	H
1	[표1]					[표2]		
2	이름	직급	기본급	보너스		직급	소속	기본급
3	홍길동	사원	1,000,000	110,000		사원	영업팀	1,000,000
4	신소영	사원	1,000,000	110,000		대리	영업팀	1,200,000
5	성준향	대리	1,200,000	130,000				
6	이몽룡	사원	1,000,000	110,000		[표3]		
7	임고운	대리	1,200,000	130,000		직급	사원	대리
8	유재성	대리	1,200,000	130,000		보너스	110,000	130,000

- [표1]의 직급[B3:B8]에 해당하는 기본급[C3:C8]을 [표2]에서 찾아 표시하시오.

▶ [C3] =VLOOKUP(B3,\$F\$3:\$H\$4,3,0)

- 자동채우기 경우 F3:H4의 참조범위가 변하면 안 되므로 <F4>키를 눌러 '\$'로 열과 행을 고정한다.
- [표2]에서 기본급이 3번째 열에 있으므로 열 번호는 3이다.
- 첫 번째 인수인 기준셀 [B3]이 문자이므로 정확한 값 0을 입력한다.

- [표1]의 직급[B3:B8]에 해당하는 보너스[D3:D8]를 [표3]에서 찾아 표시하시오.

▶ [D3] =HLOOKUP(B3,\$F\$7:\$H\$8,2,0) :

- 자동채우기 경우 F7:H8의 참조범위가 변하면 안 되므로 <F4>키를 눌러 '\$'로 열과 행을 고정한다.
- [표2]에서 기본급이 2번째 행에 있으므로 행 번호는 2이다.
- 첫 번째 인수인 기준셀 [B3]이 문자이므로 정확한 값 0을 입력한다.

- 4) =MATCH(찾을값,범위,옵션) : **범위**에서 **찾을 값**이 몇 번째 행 또는 몇 번째 열에 있는지 숫자 값으로 표시한다.

▶ **옵션** : 정확한 값을 찾으려면 0, 유사한 값을 찾으려면 1을 입력한다.(컴활2급 시험 문제로만 양기)

- 5) =INDEX(범위,행번호,열번호) : **범위** 안에서 **행번호**와 **열 번호**만큼 이동해서 값을 찾는다.



	A	B	C	D	E	F
1	[표1]					
2	이름	나이	나이 (행번호)	수강과목	과목 (열번호)	포인트
3	이몽룡	13	1	컴활2급	1	100
4	임고운	25	2	컴활2급	1	110
5	유재성	31	3	한글워드	2	150
6	이미희	22	2	한글워드	2	140
7	이성진	16	1	한글워드	2	130
8	양필숙	33	3	컴활2급	1	120
10	[포인트]					
11	나이	컴활2급	한글워드			
12	10대	100	130			
13	20대	110	140			
14	30대	120	150			

- [표1]의 나이[B3:B8]에 해당하는 행번호를 [포인트]표의 [A12:A14]에서 찾아 [C3:C8]영역에 표시하시오.

▶ [C3] =MATCH(B3,\$A\$12:\$A\$14,1)

- [A12:A14]영역에서 [B3]에 해당하는 행 번호를 찾는다.(유사한 값을 찾기 때문에 1을 입력)
- [B8]까지 자동채우기 한다.

- [표1]의 수강과목에 해당하는 열 번호를 [B11:C11]영역에서 찾아 [E3:E8]영역에 표시하시오.

▶ [E3] =MATCH(D3,\$B\$11:\$C\$11,0)

- [B11:C11]영역에서 [D3]에 해당하는 열 번호를 찾는다.(정확한 값을 찾기 때문에 0을 입력)
- [D8]까지 자동채우기 한다.

- [포인트]표의 [B12:D14]영역의 값을 [표1]의 나이(행번호)[C3:C8]와 과목(열번호)을 이용하여 찾아서 포인트[F3:F8]영역에 표시하시오.

▶ [F3] =INDEX(\$B\$12:\$C\$14,C3,E3)

- 범위 [B12:C14]에서 행번호 [C3]만큼 이동하고 열번호 [E3]만큼 이동해서 값을 찾는다.

7. 통계함수

AVERAGE, AVERAGEIF, AVERAGEIFS, COUNT, COUNTA, COUNTBLANK, COUNTIF, COUNTIFS, LARGE, MAX, MEDIAN, MIN, MODE, RANK.EQ, SMALL, STDEV, VAR

- 1) =AVERAGE(범위) : 범위 또는 인수들의 평균을 구한다.
▶ =AVERAGE(3, 4, 5) → 4
- 2) =AVERAGEIF(조건범위,조건,평균범위) : 조건범위에서 조건에 만족하는 값을 찾아서 평균범위의 평균을 구한다.
- 3) =AVERAGEIFS(평균범위,조건범위1,조건1,조건범위2,조건2...) : 조건범위1에서 조건1을 만족하고 조건범위2에서 조건2를 만족하는 평균범위의 평균을 구한다.

	A	B	C	D	E	F	G
1	이름	봉사시설	봉사시간		봉사시간 평균		4
2	이몽룡	사랑복지원	4				
3	이몽룡	애민보육원	5		사랑복지원 봉사시간 평균		3
4	성춘향	사랑복지원	3				
5	이몽룡	애민보육원	3				
6	성춘향	사랑복지원	2		이몽룡의 애민보육원 봉사시간 평균		4
7	성춘향	애민보육원	7				

- ▶ [G1] =AVERAGE(C2:C7) : 봉사시간[C2:C7]의 평균을 구한다.

▶ [G3] =AVERAGEIF(B2:B7,"사랑복지원",C2:C7)

- 봉사시설 [B2:B7]영역에서 '사랑복지원'을 찾아서 봉사시간[C2:C7]의 평균을 구한다.

▶ [G6]

=AVERAGEIFS(C2:C7,A2:A7,"이몽룡",B2:B7,"애민보육원")

- 이름[A2:A7]이 '이몽룡'이면서 봉사시설 [B2:B7]이 '애민보육원'인 봉사시간[C2:C7]의 평균을 구한다.

4) =COUNT(범위) : 범위에서 숫자, 날짜, 시간이 들어간 셀 개수를 표시한다.

▶ =COUNT(3,"2021-4-4","EBS","12:30") → 3 : 'EBS'는 문자이므로 COUNT함수가 인식하지 못한다.

5) =COUNTA(범위) : 범위에서 데이터가 있는 셀 개수를 표시한다.

▶ =COUNTA(3,"2021-4-4","EBS","12:30") → 4 : 모든 데이터의 개수를 숫자로 표시한다.

6) =COUNTBLANK(범위) : 범위에서 빈 셀의 개수를 표시한다.

7) =COUNTIF(범위,조건) → 범위에서 조건을 만족하는 셀 개수를 표시한다.

8) =COUNTIFS(조건범위1,조건1,조건범위2,조건2...) : 조건범위1에서 조건1을 만족하고 조건범위2에서 조건2를 만족하는 셀 개수를 표시한다.

	A	B	C	D	E	F	G
1	이름	봉사시설	봉사시간		봉사시간 개수		6
2	이몽룡	사랑복지원	4				
3	이몽룡	애민보육원	5		봉사시설 개수		6
4	성춘향	사랑복지원	3				
5	이몽룡	애민보육원	3		이몽룡의 개수		3
6	성춘향	사랑복지원	2				
7	성춘향	애민보육원	7		이몽룡이면서 애민보육원개수		2

- ▶ [G1] =COUNT(C2:C7) : 숫자, 날짜, 시간이 입력된 셀 개수를 구한다(문자의 개수는 셀 수 없다).

▶ [G3] =COUNTA(B2:B7) : 빈 셀을 제외한 셀 개수를 구한다.(값이 들어 있는 모든 셀의 개수를 구한다.)

▶ [G5] =COUNTIF(A2:A7,"이몽룡") : 이름 [A2:A7]에서 '이몽룡'의 셀 개수를 구한다.

▶ [G7]

=COUNTIFS(A2:A7,"이몽룡",B2:B7,"애민보육원") : [A2:A7]에서

9) =MAX(범위) : 범위에서 최대값을 구한다.



▶ =MAX(7,6,3,9,2) → 9

10) =MIN(범위) : 범위에서 **최소값**을 구한다.

▶ =MIN(7,6,3,9,2) → 2

11) =LARGE(범위,K) : 범위에서 **K번째로 큰 값**을 구한다.

12) =SMALL(범위,K) : 범위에서 **K번째로 작은 값**을 구한다

13) =RANK.EQ(기준셀,범위,옵션) : **기준셀이 범위에서 몇 등인지** 숫자로 표시한다.

▶ 옵션 : 높은 수가 1등(내림차순)이면 '0', 낮은 수가 1등(오름차순)이면 '1'을 입력한다.

14) MODE(범위) : 범위에서 **빈도수가 가장 많은 수**를 표시한다.

▶ MODE(3,2,3,2,4,3) → 3

15) MEDIAN(범위) : 범위에서 **중간값**을 구한다.

▶ =MEDIAN(3,5,11,6,12) → 6

16) =STDEV(범위) : 범위에서 **표준편차**를 구한다.

▶ =STDEV(8,2,13,9,12) → 4.32435...

17) =VAR(범위) : 범위에서 **분산**을 구한다.

▶ =VAR(8,2,13,9,12) → 18.7

	A	B	C	D	E	F	G	H
1	이름	봉사시설	봉사시간					
2	이동룡	사랑복지원	4		봉사시간의 최대값			10
3	이동룡	애민보육원	5		봉사시간의 최소값			2
4	성준향	사랑복지원	3		봉사시간에서 3번째 큰값			7
5	이동룡	애민보육원	3		봉사시간에서 3번째로 작은값			3
6	성준향	사랑복지원	2		봉사 시간에서 숫자 '4'의 순위			5
7	성준향	애민보육원	7		봉사시간에서 빈도수 많은 수			3
8	이동룡	사랑복지원	9		봉사시간의 중간값			4
9	이동룡	애민보육원	3		봉사시간의 표준편차			2.891559
10	성준향	사랑복지원	10		봉사시간의 분산			8.361111

▶ [H2] =MAX(C2:C10): 봉사시간 [C2:C10]에서 **최대값**을 구한다.

▶ [H3] =MIN(C2:C10) 봉사시간 [C2:C10]에서 **최소값**을 구한다.

▶ [H4] =LARGE(C2:C10,3): 봉사시간 [C2:C10]에서 **세 번째로 큰 값**을 구한다.

▶ [H5] =SMALL(C2:C10,3) : 봉사시간 [C2:C10]에서 **세 번째로 작은 값**을 구한다.

▶ [H6] =RANK.EQ(4,C2:C10,0) : 숫자 '4'가 봉사시간 [C2:C10]에서 **몇 등인지** 구한다(내림차순).

▶ [H7] =MODE(C2:C10) : 봉사시간 [C2:C10]에서 **빈도수가 높은 수**를 구한다.

▶ [H8] =MEDIAN(C2:C10) : 봉사시간 [C2:C10]에서 **중간값**을 구한다.

▶ [H9] =STDEV(C2:C10) : 봉사시간 [C2:C10]에서 **표준편차**를 구한다.

▶ [H10] =VAR(C2:C10) : 봉사시간 [C2:C10]에서 **분산**을 구한다.